JOURNAL OF APPLIED POLYMER SCIENCE VOL. 17, PP. 1449–1454 (1973)

Intrinsic Viscosity Calculated by Computer

B. T. HOFREITER, J. O. ERNST, and W. L. WILLIAMS,
Northern Regional Research Laboratory, Agricultural Research Service, U.S. Department of Agriculture, Peoria, Illinois 61604

Synopsis

A computer program has been devised for routine use which processes raw data such as polymer sample weight, moisture content, solution volume, and viscometer flow times to calculate a least-squares-derived intrinsic viscosity. In addition to eliminating errors inherent in graphic solutions and freeing technicians from tedious calculation, the computer output provides the 95% confidence interval of the intrinsic viscosity.

INTRODUCTION

The well-known calculations required in determining intrinsic viscosities of polysaccharides and other polymers are tedious and subject to human error in curve fitting. Usually, intrinsic viscosity is determined by a graphic solution in which the value of reduced viscosity is extrapolated to zero concentration on a reduced viscosity–concentration plot. Furthermore, the graphic solution does not facilitate statistical treatment of precision. Precision of an intrinsic viscosity is important in determining real effects of physical and chemical treatments on polymer molecular weight or structure as inferred from viscosity changes.

To obviate these difficulties, a computer program was prepared for rapidly calculating intrinsic viscosity from such raw data as grams of polymer, polymer moisture, dilution volume, and viscometer flow times of solvent and solutions. The essential features of computer processing intrinsic viscosity data are:

1. A raw data Intrinsic Viscosity Input Form (Fig. 1) that is completed by the technician in the laboratory and from which data cards are keypunched.

2. Output from computer (Fig. 2) that includes: (a) a one-line, 80-character description of sample, solvents, operator, and date; (b) raw data printout; (c) summary of reduced viscosity (Nsp/C) and corresponding concentration (C); (d) intrinsic viscosity to three decimal places; (e) 95% confidence interval (C.I.) of intrinsic viscosity value.

3. A graphic representation (Fig. 3) of the data points to enable ready identification of any outliers. This optional feature requires some additional programming (not described here) and a computer-plotter attachment. However, as an alternative to the plotter, whenever the C.I. ex-



Fig. 1. Raw data input form showing sample data from which Figures 2 and 3 were obtained. Data cards are keypunched directly from this form.

```
                    INTRINSIC VISCOSITY

COMMERCIAL AMYLOPECTIN (9517-28), VISCOMETER A21, DMSO SOLVENT, 11-16-71, B.T.M.

 GRAMS SAMPLE  0.0904 VOLUMETRIC  50.0 ML   MOISTURE  8.82 PERCENT

 FLOW TIMES, SEC.

   AVERAGE            DATA
   166.6     SOLVENT 166.6 166.6 166.6
   199.5     SOLN C1 199.5 199.5 199.5
   181.9     SOLN C2 181.9 182.0 181.9 181.9
   176.6     SOLN C3 176.6 176.6 176.8 176.7
   174.0     SOLN C4 173.9 174.1 174.1

SUMMARY
            DILN          VISCOSITY
  CONC.PCT  FACTOR    REL    SPECIFIC REDUCED

   0.1648     1.0   1.1974   0.1974   1.1979
   0.0824     2.0   1.0919   0.0919   1.1159
   0.0549     3.0   1.0604   0.0604   1.1003
   0.0412     4.0   1.0446   0.0446   1.0826

 INTRINSIC VISCOSITY=   1.045  DL/GRAM      C I = 0.021
```

Fig. 2. Computer output from intrinsic viscosity program.

ceeds an established norm, Nsp/C versus C should be plotted from values calculated in the computer output (Fig. 2). In this manner any disproportionate error contributed by a single data point set can be determined.

Outputs can be bound to give a permanent laboratory notebook record of raw and derived data.

PROGRAM OPERATION

The mechanics of the program use are illustrated by the example data in Figure 1:

1. Preparation of Intrinsic Viscosity Input Form (Fig. 1).

Card 1:

Column 1. Total number of polymer dilutions to be run.

Column 2. Code number indicating whether or not a different viscometer is used for each polymer concentration.

Columns 3–8. Weight in grams, "as is," of polymer.



Fig. 3. Plotter output showing scatter of experimental points and least-squares-derived straight line.

Columns 9–13. Volume, in milliliters, to which the polymer is diluted in the stock solution (C_1).

Columns 14–18. Per cent moisture content of polymer.

However, if one begins with a polymer solution of known concentration, columns 3–18 (Card 1) may be filled assuming a a 100-ml volume and zero moisture.

Card 2:

Columns 1–80. All references and comments.

Card 3:

Columns 1–50 (by groups of five). Solvent flow times, to the nearest tenth second. If an average value for the viscometer is already known, it should be recorded in columns 1–5.

```
// JOB
// FOR
* ONE WORD INTEGERS
* EXTENDED PRECISION
* IOCS ( CARD , 1132 PRINTER )
      DIMENSION FLOWS(10),FLOW(10),DIFLN(9),RLVIS(9),SPVIS(9)
      DIMENSION X(9),Y(9),ST(7)
      DIMENSION IOUTC(80)
      DATA ST(7),ST(6),ST(5),ST(4),ST(3),ST(2),ST(1)
     1/ 2.3646,2.4469,2.5706,2.7764,3.1825,4.3027,12.706 /
  200 READ(2,205)NOSLN,NOSET,GRSOL,VOLML,PCTM
  205 FORMAT(2I1,F6.4,F5.2,F5.3)
      IF(NOSLN)295,295,210
  295 CALL EXIT
  210 ACONC=(GRSOL-PCTM*GRSOL/100.)*100./VOLML
      WRITE(3,208)
  208 FORMAT('1'20X'INTRINSIC VISCOSITY')
      READ(2,222)IOUTC
  222 FORMAT(80A1)
      WRITE(3,223)IOUTC
  223 FORMAT('0'80A1)
      WRITE(3,212)GRSOL,VOLML,PCTM
  212 FORMAT('0 GRAMS SAMPLE'F8.4,' VOLUMETRIC'F6.1,' ML   MOISTURE'
     1F6.2' PERCENT'/'0 FLOW TIMES, SEC.'/'0   AVERAGE'11X'DATA')
      DO 250 J=1,NOSLN
      IF(NOSET)214,211,214
  211 IF(J-1)227,214,227
  214 READ(2,215)FLOWS
  215 FORMAT(10F5.1)
      NO=0
      TFLOS=0.
      DO 220 I=1,10
      IF(FLOWS(I))220,220,218
  218 NO=NO+1
      TFLOS=TFLOS+FLOWS(I)
  220 CONTINUE
      TFLOS=TFLOS/NO
      WRITE(3,225)TFLOS,(FLOWS(I),I=1,NO)
  225 FORMAT(3XF6.1,4X'SOLVENT'10F6.1)
  227 READ(2,232)DIFLN(J),FLOW
  232 FORMAT(F5.2,10F5.1)
      NO=0
      TFLO=0.
      DO 240 I=1,10
      IF(FLOW(I))240,240,235
  235 NO=NO+1
      TFLO=TFLO+FLOW(I)
  240 CONTINUE
      TFLO=TFLO/NO
      WRITE(3,230)TFLO,J,(FLOW(I),I=1,NO)
  230 FORMAT(3XF6.1,4X'SOLN C'I1,10F6.1)
```

Fig. 4 (*continued*)

```
      X(J)=ACONC/DIFLN(J)
      RLVIS(J)=TFLO/TFLOS
      SPVIS(J)=RLVIS(J)-1.
      Y(J)=SPVIS(J)/X(J)
  250 CONTINUE
      WRITE(3,255)
  255 FORMAT ('0SUMMARY'/   14X'DILN'8X'VISCOSITY'
     1/3X'CONC,PCT  FACTOR    REL   SPECIFIC REDUCED'/1X)
      WRITE(3,249)(X(J),DIFLN(J),RLVIS(J),SPVIS(J),Y(J),J=1,NOSLN)
  249 FORMAT (1XF9.4,F8.1,F10.4,F8.4,F9.4)
      N=NOSLN
      IF(N-1)200,200,3
    3 EN=N
      SIGX=0.0
      SIGY=0.0
      SIGXX=0.0
      SIGXY=0.0
      DO 101 I=1,N
      SIGX=SIGX+X(I)
      SIGY=SIGY+Y(I)
      SIGXX=SIGXX+X(I)*X(I)
      SIGXY=SIGXY+X(I)*Y(I)
  101 CONTINUE
      A=(SIGY*SIGXX-SIGX*SIGXY)/(EN*SIGXX-SIGX**2)
      B=(EN*SIGXY-SIGX*SIGY)/(EN*SIGXX-SIGX**2)
      ERA=0.
      DO 70 I=1,N
      YEA=X(I)*B+A
      YERA=Y(I)-YEA
      ERA=ERA+YERA*YERA
   70 CONTINUE
      NDF=N-2
      DF=NDF
      IF(NDF)340,340,342
  340 CI=0.
      GO TO 348
  342 CI=SQRT(ERA/DF*(1./EN+SIGX**2/(EN**2*SIGXX-EN*SIGX**2)))
      CI=CI*ST(NDF)
  348 WRITE(3,260)A,CI
  260 FORMAT ('0  INTRINSIC VISCOSITY='F8.3'  DL/GRAM'5X'C I ='F6.3)
      GO TO 200
      END
// XEQ
40  0904 50    882
COMMERCIAL AMYLOPECTIN (9517-28), VISCOMETER A21, DMSO SOLVENT, 11-16-71, B.T.M.
 1666 1666 1666
  1    1995 1995 1995
  2    1819 1820 1819 1819
  3    1766 1766 1768 1767
  4    1739 1741 1741
0
```

Fig. 4. Computer program in Fortran IV for calculation of intrinsic viscosity.

Card 4:

Columns 1–5. Dilution factor; e.g., a dilution factor of the stock solution is 1.0 and a 1:1 dilution of the stock solution is a dilution factor of 2.

Columns 6–55 (by groups of five). C_1 flow time (initial concentration).

At this point, if only one viscometer is used throughout as in the sample data (Fig. 1), there will be only one solvent flow time card. However, when there is a different viscometer for each polymer concentration, the solvent flow time card is paired with the corresponding solution flow time card according to the card sequence shown in Figure 1. Any number of polymer solutions up to a total of nine may be run with the present program.

2. Data cards keypunched from the Intrinsic Viscosity Input Form, placed in the order shown in Figure 1, follow the program cards in the computer processing.

CALCULATIONS

The program computes from experimental data, by the method of least squares, the slope and intercept of eq. (1):

$$X = (M)\ (C) + \eta \tag{1}$$

where X = Nsp/C, M = slope, C = concentration, g/100 ml; and η = Intercept (intrinsic viscosity), dl/g.

Presumably, regression of reduced viscosity on polymer concentration is linear. Although our Fortran program would be inadequate for non-swamped polyelectrolytes, it could be modified to process quadratic regression and, thereby, extrapolate reduced viscosity more accurately to infinite dilution.

C.I. is at the 95% level and is defined by eq. (2):

$$\text{C.I.} = t_{\alpha,N-2} \cdot S_{Y \cdot X} \sqrt{\frac{1}{N} + \frac{\bar{X}^2}{\Sigma(X_i - \bar{X})^2}} \tag{2}$$

where $S_{Y \cdot X}$ = standard error of estimate of the regression of reduced viscosity on concentration; $t_{\alpha,N-2}$ = statistical "t" evaluated for $N-2$ degrees of freedom where α is the probability of a larger value of t (sign ignored) and $100(1-\alpha)$ is the level of confidence; N = number of experimental points; $\bar{X}$ = average value of reduced viscosity; and X_i = values of i numbers of reduced viscosity.

The meaning of C.I. at the 95% level is: The true or population intrinsic viscosity will be included within the confidence limits of 95% of the viscosity determinations. It is important to know that the t value in eq. (2) has values for N = 3, 4, 5, and 6 observations of $t_{\alpha,N-2}$ = 12.71, 4.30, 3.18, and 2.78, respectively. Consequently, C.I. is reduced about one third in going from three to four points because two degrees of freedom are lost in the regression, which for three observations leaves only one degree of freedom.

The computer program, which is in Fortran IV, is given in Figure 4.

The mention of firm names or trade products does not imply that they are endorsed or recommended by the U.S. Department of Agriculture over other firms or similar products not mentioned.

Received September 29, 1972